

Math 2250 Maple Project 2

Spring 2001 revision of David Eyre's project (2000 version).

There are six (6) problems in this project. You are expected to answer the questions A, B, C, ... associated with each problem. The original worksheet "project2.mws" is a template for the solution; you must fill in the code and all comments. Sample code can be copied with the mouse. Please use pencil

freely to annotate the worksheet and to clarify any code or figure presented herein.

The problem headers:

2.1. OVERDAMPED FREE OSCILLATIONS.

The first four problems study the simplest linear model for  $x(t)$ .

2.2. UNDERDAMPED FREE OSCILLATIONS.

2.3. UNDAMPED FORCED OSCILLATIONS ( $c=0$ ).

2.4. DAMPED FORCED OSCILLATIONS ( $c>0$ ).

2.5. LARGE SUSTAINED OSCILLATIONS.

This is the second, nonlinear model.

2.6. MCKENNA NON-HOOKES LAW CABLE MODEL. This is the third, nonlinear model.

FREE OSCILLATIONS. Consider the general problem of free linear oscillations

$$m x'' + c x' + k x = 0,$$
$$x(0) = x_0, \quad x'(0) = v_0,$$

where  $m$ ,  $c$  and  $k$  are non-negative constants. The symbols  $x_0$  and  $v_0$  are the initial position and initial velocity, respectively.

2.1. PROBLEM (OVERDAMPED FREE OSCILLATIONS)

- A. Let  $m=k=1$ . Suggest a value for parameter  $c > 0$  so that the free oscillations are overdamped.
- B. Use  $x(0)=1$  and  $x'(0)=-2$  for the initial conditions and Maple's "dsolve" to find the explicit real solution  $x(t)$ . Plot the solution  $x(t)$  for  $t=0$  to  $t=5$  using Maple's "plot" command.
- C. Suggest new values for  $x(0)$  and  $x'(0)$  such that the solution is non-negative, increases near  $t=0$  but eventually decreases with limit zero at  $t=\infty$ . Find the explicit solution and plot it for  $t=0$  to  $t=5$ .
- D. Suggest new values for  $x(0)$  and  $x'(0)$  such that the solution changes sign exactly once on  $t>0$ , then decreases to zero at  $t=\infty$ . Find the explicit solution and plot it for  $t=0$  to  $t=5$ .

EXAMPLE(Wrong parameters! Change it!)

```
de:=diff(x(t),t,t)-diff(x(t),t)+x(t)=0:
```

```
ic:=x(0)=1,D(x)(0)= -2:
```

```
dsolve({de,ic},x(t),method=laplace);
```

```
X:=unapply(rhs(%),t):
```

```
plot(X(t),t=0..5);
```


```
>
```

```

> #2.1-A and 2.1-B
de:=diff(x(t),t,t)+2*diff(x(t),t)+x(t)=0:
ic:=x(0)=1,D(x)(0)= -2:
dsolve({de,ic},x(t),method=laplace);
X:=unapply(rhs(%),t):
plot(X(t),t=0..5);

```

$$x(t) = -t e^{(-t)} + e^{(-t)}$$




```

> #2.1-C
ic:=x(0)=1,D(x)(0)= 2:
dsolve({de,ic},x(t),method=laplace);
X:=unapply(rhs(%),t):
plot(X(t),t=0..5);


```

$$x(t) = 3 t e^{(-t)} + e^{(-t)}$$



```
> #2.1-D  
ic:=x(0)= -1,D(x)(0)= 2:  
dsolve({de,ic},x(t),method=laplace);  
X:=unapply(rhs(%),t):  
plot(X(t),t=0..5);
```

$$x(t) = t e^{(-t)} - e^{(-t)}$$




## 2.2. PROBLEM (UNDERDAMPED FREE OSCILLATIONS)

A. Let  $m=1, k=5$ . Find a parameter value  $c > 0$  so that the solution  $x(t)$  changes sign infinitely many times and decays to zero at  $t=\infty$ . Plot the solution for initial values  $x(0)=0, x'(0)=1$  on  $t=0$  to  $t=5$ .

B. Estimate from the graph the pseudoperiod of the solution.

```
> #2.2-A Choose m=1, c=2, k=5.
de:=diff(x(t),t,t)+2*diff(x(t),t)+5*x(t)=0:
ic:=x(0)=0,D(x)(0)=1:
dsolve({de,ic},x(t),method=laplace);
X:=unapply(rhs(%),t):
plot(X(t),t=0..5);
```

$$x(t) = \frac{1}{2} e^{-t} \sin(2t)$$



[ > #2.2-B The pseudoperiod is  $\pi$ .

[ >

### FORCED LINEAR OSCILLATIONS.

Consider the forced problem

$$m x'' + c x' + k x = F \cos(\omega t),$$

$$x(0) = x_0, \quad x'(0) = v_0,$$

where  $m, c, k, \omega$  are non-negative constants and  $F$  is a nonzero constant.

The symbols  $x_0$  and  $v_0$  are the initial position and initial velocity, respectively.

### 2.3. PROBLEM (UNDAMPED FORCED OSCILLATIONS ( $c=0$ ))

A. Choose  $m=1, c=0, k=4, \omega=6$ , so that the forcing frequency  $\omega$  is 3 times larger than the natural frequency  $\omega_0=2$ . Let  $F=1, x_0=0, v_0=0$ . Solve for  $x(t)$  using `dsolve()`. Plot the solution  $x(t)$  on a suitable interval in order to show the global behavior of the solution  $x(t)$ .

B. The solution  $x(t)$  is the sum of two functions, one of period  $2\pi/\omega$  and the other of period  $2\pi/\omega_0$ .

Find the period of  $x(t)$  by examining the graph and the equation for  $x(t)$ .

C. Let  $c=0, m=1, k=4, F=10$ . Suggest a value for the forcing frequency  $\omega$  so that the oscillations exhibit


resonance. Show resonant behavior on a graph using initial values  $x(0)=0, x'(0)=0$ .

```

>
> #2.3-A
de:=diff(x(t),t,t)+4*x(t)=cos(6*t):
ic:=x(0)=0,D(x)(0)=0:
dsolve({de,ic},x(t),method=laplace);
X:=unapply(rhs(%),t):
plot(X(t),t=0..5);

```

$$x(t) = \frac{1}{32} \cos(2t) - \frac{1}{32} \cos(6t)$$



```

> #2.3-B
period:=Pi;

```


$$period := \pi$$

```

> #2.3-C
de:=diff(x(t),t,t)+4*x(t)=10*cos(2*t):
ic:=x(0)=0,D(x)(0)=0:
dsolve({de,ic},x(t),method=laplace);
X:=unapply(rhs(%),t):
plot(X(t),t=0..20);

```

$$x(t) = \frac{5}{2} t \sin(2t)$$



>

#### 2.4. PROBLEM (DAMPED FORCED OSCILLATIONS ( $c > 0$ ))

- Choose parameter values  $m=1$ ,  $c=2$ ,  $k=25$ ,  $F=10$ ,  $w=1$ ,  $x_0=0$ ,  $v_0=0$ . Solve for  $x(t)$  and plot the solution on  $t=0$  to  $t=10$ .
- Extract from the solution  $x(t)$  the steady-state solution  $x_{ss}(t)$ . Plot it on  $t=0$  to  $t=10$ .
- Choose  $F = 10$ ,  $m=1$ ,  $k = 25$ , and consider  $c=2$ ,  $c=1$ ,  $c=1/2$ . Compute the amplitude function  $C(w)$  of  $x_{ss}(t)$  [page 346] for these three equations, then plot for  $w=0$  to  $w=20$  the three amplitude graphs on a single set of axes.
- Print the natural frequency  $w$  and the maximum amplitude  $C(w)$  for each case  $c=2$ ,  $c=1$ ,  $c=1/2$ . The three data pairs should show that  $\max C(w)$  becomes larger as  $c$  tends to zero.

EXAMPLE(Beware! Wrong values!)

$F:=5$ :  $m:=1$ :  $k:=25$ :  $c:=c$ :  $w:=w$ :

$C:=(w,c) \rightarrow F/\sqrt{(k-m*w*w)^2+(c*w)^2}$ :

$\text{plot}(\{C(w,4),C(w,3),C(w,2)\},w=0..15,\text{color}=\text{black});$

>

> #2.4-A


$de:=\text{diff}(x(t),t,t)+2*\text{diff}(x(t),t)+25*x(t)=10*\cos(t):$

$ic:=x(0)=0,D(x)(0)=0:$

$dsolve(\{de,ic\},x(t),\text{method}=\text{laplace});$


```
X:=unapply(rhs(%),t):  
plot(X(t),t=0..10);
```

$$x(t) = -\frac{12}{29} e^{(-t)} \cos(2\sqrt{6}t) - \frac{13}{348} \sqrt{6} e^{(-t)} \sin(2\sqrt{6}t) + \frac{12}{29} \cos(t) + \frac{1}{29} \sin(t)$$




```
> #2.4-B  
Xss:=t->12/29*cos(t)+1/29*sin(t):  
plot(Xss(t),t=0..10);
```





```
> #2.4-C  
F:=10: m:=1: k:=25: c:='c': w:='w':  
C:=(w,c)->F/sqrt((k-m*w*w)^2+(c*w)^2):  
plot({C(w,2),C(w,1),C(w,1/2)},w=0..20,color=black);
```



```
> #2.4-D
Cmax:=4.00; Cmax:=2.00; Cmax:=1.01;
W:=5.01; W:=4.95; W:=4.83;
```

```
Cmax := 4.00
```

```
Cmax := 2.00
```

```
Cmax := 1.01
```

```
W := 5.01
```

```
W := 4.95
```

```
W := 4.83
```

```
[ >
```

```
[ >
```

**NONLINEAR MODEL WITH GEOMETRY INCLUDED.**

Consider the nonlinear, forced, damped oscillator equation for torsional motion, with bridge geometry included,

$$mx'' + c x' + k \sin(x)\cos(x) = F \cos wt ,$$

$$x(0) = x_0, \quad x'(0) = v_0$$

and its corresponding linearized equation

$$mx'' + c x' + k x = F \cos wt ,$$

$$x(0) = x_0, \quad x'(0) = v_0.$$

Maple code used to solve and plot the solutions appears below.

```

m:=1: F := 10: w := 5: k := 25: c:= 7:
deNonLinear:= m*diff(x(t),t,t) + c*diff(x(t),t) + k*sin(x(t))*cos(x(t)) = F*cos(w*t):
deLinear:= m*diff(x(t),t,t) + c*diff(x(t),t) + k*x(t) = F*cos(w*t):
with(DEtools): opts:=stepsize=0.1:
DEplot(deNonLinear,x(t),t=0..2,[[x(0)=0,D(x)(0)=0]],opts,title='NonLinear');
DEplot(deLinear,x(t),t=0..2,[[x(0)=0,D(x)(0)=0]],opts,title='Linear');

```

## 2.5. PROBLEM (LARGE SUSTAINED OSCILLATIONS)

- A. Set the parameters in the linear and nonlinear equations equal to the following:  
 $m=1$ ,  $c = 0.05$ ,  $k = 2.4$ ,  $w = 1.2$ ,  $F = 0.06$ ,  $x_0=0$ ,  $v_0=0$ . Solve and plot the solutions of the linear and nonlinear equations from  $t=150$  to  $t=250$ .
- B. Set the parameters in the linear and nonlinear equations equal to the following:  
 $m=1$ ,  $c = 0.05$ ,  $k = 2.4$ ,  $w = 1.2$ ,  $F = 0.06$ ,  $x_0=1.2$ ,  $v_0=0$ . Solve and plot the solutions of the linear and nonlinear equations from  $t=200$  to  $t=300$ .
- C. The plots made in A and B represent the steady-state solutions of the two equations under two different sets of initial conditions. Argue in a sentence why the two linear plots have to be identical, based upon the superposition formula  $x(t)=x_h(t)+x_{ss}(t)$ , even though the homogeneous solution  $x_h(t)$  is different for the two plots. Finally, determine the ratio of the apparent amplitudes (a number  $> 1$ ) for the nonlinear plots and explain why "large sustained oscillations" is an appropriate description of the nonlinear steady-state behavior.

>

> #2.5-A



```

m:=1: F := 0.06: w := 1.2: k := 2.4: c:= 0.05:
opts:=stepsize=0.1:
deNonLinear:= m*diff(x(t),t,t) + c*diff(x(t),t) +
k*sin(x(t))*cos(x(t)) = F*cos(w*t):
deLinear:= m*diff(x(t),t,t) + c*diff(x(t),t) + k*x(t) =
F*cos(w*t):
with(DEtools):

DEplot(deNonLinear,x(t),t=150..250,[[x(0)=0,D(x)(0)=0]],opts,title
='NonLinear');



DEplot(deLinear,x(t),t=150..250,[[x(0)=0,D(x)(0)=0]],opts,title='L
inear');

```



```
> #2.5-B  
DEplot(deNonLinear,x(t),t=200..300,[[x(0)=1.2,D(x)(0)=0]],opts,title='NonLinear');
```

```
DEplot(deLinear,x(t),t=200..300,[[x(0)=1.2,D(x)(0)=0]],opts,title=
'Linear');
```



> #2.5-C

```

# On the interval, xh(t) approximately zero, so the two linear
# plots must look the same.
# From the nonlinear graphs, the amplitude ratio is
0.95/0.06=15.83.
# The second graph shows after 200 seconds torsional oscillations
# close to one radian = 57.3 degrees, hence a small disturbance
x(0)=1.2
# produces large sustained oscillations.
0.95/0.06;evalf(180/Pi);

15.83333333
57.29577950

```

>

## MCKENNA'S NON-HOOKE'S LAW CABLE MODEL FOR THE TACOMA NARROWS BRIDGE

The model of McKenna studies the bridge with a nonlinear, forced, damped oscillator equation for torsional motion that accounts for the non-Hooke's law cables coupled to the equations for vertical motion. The equations in this case couple the torsional motion with the vertical motion. The equations are:

$$\begin{aligned}
 x'' + c x' - k G(x,y) &= F \sin wt, & x(0) &= x_0, & x'(0) &= x_1, \\
 y'' + c y' + (k/3) H(x,y) &= g, & y(0) &= y_0, & y'(0) &= y_1,
 \end{aligned}$$

where  $x(t)$  is the torsional motion and  $y(t)$  is the vertical motion. The functions  $G(x,y)$  and  $H(x,y)$  are the models of the force generated by the cable when it is contracted and stretched. Below is sample code for writing the differential equations and for plotting the solutions. It is ready to copy with the mouse.

```

with(DEtools):
w := 1.3: F := 0.05: f(t) := F*sin(w*t):
c := 0.01: k1 := 0.2: k2 := 0.4: g := 9.8: L := 6:
STEP:=x->piecewise(x<0,0,1):
fp(t) := y(t)+(L*sin(x(t))):
fm(t) := y(t)-(L*sin(x(t))):
Sm(t) := STEP(fm(t))*fm(t):
Sp(t) := STEP(fp(t))*fp(t):
sys := {
  diff(x(t),t,t) + c*diff(x(t),t) - k1*cos(x(t))*(Sm(t)-Sp(t))=f(t),
  diff(y(t),t,t) + c*diff(y(t),t) + k2*(Sm(t)+Sp(t)) = g}:
ic := [[x(0)=0, D(x)(0)=0, y(0)=27.25, D(y)(0)=0]]:
vars:=[x(t),y(t)]:
opts:=stepsize=0.1:
DEplot(sys,vars,t=0..300,ic,opts,scene=[t,x]);

```

The amazing thing that happens in this simulation is that the large vertical oscillations take all the tension out of the springs and they induce large torsional oscillations.

### 2.6. PROBLEM. ( MCKENNA'S NON-HOOKE'S LAW CABLE MODEL)

- A. TORSIONAL OSCILLATION PLOT. Get the sample code above to produce the plot of  $x(t)$  [that's what scene=[t,x] means]. Estimate the number of degrees the roadway oscillates based on the plot; recall that  $x$  in the plot is reported in radians.
- B. VERTICAL OSCILLATION PLOT. Modify the DEplot code to scene=[t,y] and plot the oscillation  $y(t)$  on  $t=0$  to  $t=300$ . The plot is supposed to show 30-foot vertical oscillations that dampen to 7-foot vertical oscillations after 300 seconds. Imagine the auto in the Tacoma Narrows film clip undergoing 30-foot vertical excursions!

```


>
> #2.6-A
with(DEtools):

w := 1.3:  F := 0.05:  f(t) := F*sin(w*t):

c := 0.01:  k1 := 0.2:  k2 := 0.4:  g := 9.8:  L := 6:

STEP:=x->piecewise(x<0,0,1):
fp(t) := y(t)+(L*sin(x(t))):
fm(t) := y(t)-(L*sin(x(t))):
Sm(t) := STEP(fm(t))*fm(t):
Sp(t) := STEP(fp(t))*fp(t):
sys := {
    diff(x(t),t,t) + c*diff(x(t),t) -
k1*cos(x(t))*(Sm(t)-Sp(t))=f(t),
    diff(y(t),t,t) + c*diff(y(t),t) + k2*(Sm(t)+Sp(t)) = g}:
ic := [[x(0)=0, D(x)(0)=0, y(0)=27.25, D(y)(0)=0]]:
vars:=[x(t),y(t)]:
opts:=stepsize=0.1:
DEplot(sys,vars,t=0..300,ic,opts,scene=[t,x]);

```




```
> #2.6-A Answer for degrees  
Degrees:=evalf(0.97*180/Pi);
```

*Degrees := 55.57690611*

```
> #2.6-B  
DEplot(sys,vars,t=0..300,ic,opts,scene=[t,y]);
```





[ >  
[ >  
[ >